Affine Abbildungen Drehungen, Spiegelungen...

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 3.1.1, Okt. 05 Update 27.10.05
Web: http://haftendorn.uni-lueneburg.de www.mathematik-verstehen.de
###
Dieses Notebook eignet sich für viele Beispiele. "Grün" muss angepasst werden. (3 Stellen) ###

Die Spalten der Abbildungsmatrix sind Bilder von $\vec{e}_1 = \begin{pmatrix} 1 \\ 0 \end{pmatrix}$ $\vec{e}_1 = \begin{pmatrix} 0 \\ 1 \end{pmatrix}$

```
Eh:=matrix([[1,0],[0,1]])
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

Drehung um Ursprung mit Winkel phi

$$\vec{p}\,' = \begin{pmatrix} \cos(\varphi) & -\sin(\varphi) \\ \sin(\varphi) & \cos(\varphi) \end{pmatrix} \cdot \vec{p}$$

```
Dr:=phi-> matrix([[cos(phi), -sin(phi)], [sin(phi), cos(phi)]])
Dr(phi)
```

$$\begin{pmatrix} \cos(\mathrm{phi}) & -\sin(\mathrm{phi}) \\ \sin(\mathrm{phi}) & \cos(\mathrm{phi}) \end{pmatrix}$$

Spiegelung an der x-Achse

```
Spx:=matrix(
[[1,0],[0,-1]]);
```

$$\begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$$

Spiegelung an der y-Achse

```
Spy:=matrix([[-1,0],[0,1]]);
```

$$\begin{pmatrix} -1 & 0 \\ 0 & 1 \end{pmatrix}$$

Spiegelung an der Ursprungs-Geraden y=m x
gw= Steigungswinkel in RAD.

```
gw:=arctan(m);
```

$$-\arctan\left(\frac{1}{2}\right)$$

```
Spm:=Dr(gw)*Spx*Dr(-gw)
```

$$\begin{pmatrix} \frac{3}{5} & -\frac{4}{5} \\ -\frac{4}{5} & -\frac{3}{5} \end{pmatrix}$$

$$Spm := \frac{1}{m^2+1} \begin{pmatrix} 1-m^2 & 2m \\ 2m & m^2-1 \end{pmatrix}$$

$$Spm := \frac{1}{m^2+1}\begin{pmatrix} 1-m^2 & 2m \\ 2m & m^2-1 \end{pmatrix}$$

Von Hand ginge das so:

```
Dr(g)*Spx*Dr(-g);
```

$$\begin{pmatrix} \cos(g)^2-\sin(g)^2 & 2\cdot\cos(g)\cdot\sin(g) \\ 2\cdot\cos(g)\cdot\sin(g) & \sin(g)^2-\cos(g)^2 \end{pmatrix}$$

Additionstheoreme

```
expand(sin(2*t)),expand(cos(2*t));
```

$$2\cdot\cos(t)\cdot\sin(t),\ \cos(t)^2-\sin(t)^2$$

```
combine(cos(t)^2-sin(t)^2,sincos);
combine(2*cos(t)*sin(t),sincos);
```

$$\cos(2\cdot t)$$

$$\sin(2\cdot t)$$

```
rewrite(cos(2*g),tan);
rewrite(sin(2*g),tan)
```

$$-\frac{\tan(g)^2-1}{\tan(g)^2+1}$$

$$\frac{2\cdot\tan(g)}{\tan(g)^2+1}$$

mit tan(g)=m erhält man dann die angegebene Matrix

----

gw= Steigungswinkel in RAD.

```
Spmw:=matrix([[cos(2*gw),sin(2*gw)],[sin(2*gw),-cos(2*gw)]])
```

$$\begin{pmatrix} \cos(2\cdot\arctan(\frac{1}{2})) & -\sin(2\cdot\arctan(\frac{1}{2})) \\ -\sin(2\cdot\arctan(\frac{1}{2})) & -\cos(2\cdot\arctan(\frac{1}{2})) \end{pmatrix}$$

Spiegelung an der Geraden y=m x+b

```
bvec:=matrix([0,b])
```

$$\begin{pmatrix} 0 \\ b \end{pmatrix}$$

$$\vec{p}\,' = Spm\cdot\left(\vec{p}-\vec{b}\right)+\vec{b}$$

$$\vec{p}\,' = Spy \cdot \left(\vec{p} - \vec{a}\right) + \vec{a}$$

Zentrische Streckung von Ursprung aus mit Faktor k

```
Zk:=k*Eh;
```

$$\begin{pmatrix} k & 0 \\ 0 & k \end{pmatrix}$$

Scherung mit y-Achse als Scherachse , Steigungswinkel von Steigung m als Scherwinkel

```
Scher:=matrix([[1,0],[-m,1]])
```

$$\begin{pmatrix} 1 & 0 \\ \frac{1}{2} & 1 \end{pmatrix}$$

```
fkt:=matrix([x,f(x)])
```

$$\begin{pmatrix} x \\ f(x) \end{pmatrix}$$

```
Scher*fkt
```

$$\begin{pmatrix} x \\ \frac{x}{2} + f(x) \end{pmatrix}$$

Hier sieht man, dass solche Scherungen Additionen von Ursprungsgeragen sind.

++

Definition eines Urbildes , hier eintragen

```
myUr:=[[2,0],[2,1],[1,1],[1,3],[0,3],[0,0],[2,0],[1,1/2]]:
myUrM:=linalg::transpose(matrix(myUr))
```

$$\begin{pmatrix} 2 & 2 & 1 & 1 & 0 & 0 & 2 & 1 \\ 0 & 1 & 1 & 3 & 3 & 0 & 0 & \frac{1}{2} \end{pmatrix}$$

```
npk:=nops(myUr):   //Anzahl der Punkte
```

++

Verschiebung

Translationsvektor hier eintragen Weiterer Hilfsvektor a hier eintragen

```
trans:=matrix([0,0]);
avec:=matrix([a,b]);
```

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix}$$

$$\begin{pmatrix} a \\ b \end{pmatrix}$$

$$\begin{pmatrix} a \\ b \end{pmatrix}$$

Aufblasen der Translationsvektoren zu Matrizen (als Funktion)

```
tm:=tr->linalg::transpose(matrix([[tr[1],tr[2]] $ npk])):
tm(trans);
tm(avec);
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} a & a & a & a & a & a & a & a \\ b & b & b & b & b & b & b & b \end{pmatrix}$$

Auswahl, welche Abbildung gezeichnet werden soll. Aktiviere durch Anpassen und Fortnehmen der //

```
//A:=Dr(PI/3);// hier Drehwinkel eintragen
```

```
//A:=Spx;
```

```
A:=Spm; m:=-1/2 // hier m eintragen
```

$$\begin{pmatrix} \frac{3}{5} & -\frac{4}{5} \\ -\frac{4}{5} & -\frac{3}{5} \end{pmatrix}$$

$$-\frac{1}{2}$$

```
spiegelgerade:=plot::Function2d(m*x,x=-3..3):
```

```
urbild:=plot::Polygon2d(myUr,
LineWidth=1, LineColor=RGB::Green, Scaling=Constrained):
e1s:=plot::Arrow2d([0,0],[A[1,1],A[2,1]]):
e2s:=plot::Arrow2d([0,0],[A[1,2],A[2,2]],LineColor=RGB::Black):
transl:=plot::Arrow2d([0,0],trans,LineColor=RGB::Magenta):
plot(urbild,e1s,e2s,transl,spiegelgerade,GridVisible=TRUE, View:
```



Transponieren, damit die Spalten die Urbild-Punkte sind:

```
myUrM:=linalg::transpose(matrix(myUr))
```

$$\begin{pmatrix} 2 & 2 & 1 & 1 & 0 & 0 & 2 & 1 \\ 0 & 1 & 1 & 3 & 3 & 0 & 0 & \frac{1}{2} \end{pmatrix}$$

```
myBM:=A*myUrM +tm(trans)    //die Bildpunkte
```

$$\begin{pmatrix} \frac{6}{5} & \frac{2}{5} & -\frac{1}{5} & -\frac{9}{5} & -\frac{12}{5} & 0 & \frac{6}{5} & \frac{1}{5} \\ -\frac{8}{5} & -\frac{11}{5} & -\frac{7}{5} & -\frac{13}{5} & -\frac{9}{5} & 0 & -\frac{8}{5} & -\frac{11}{10} \end{pmatrix}$$

```
myB:=[[myBM[1,j],myBM[2,j]] $ j=1..npk]:
bild:=plot::Polygon2d(myB,
LineWidth=1, LineColor=RGB::Red):
plot(urbild,bild,e1s,e2s,transl,spiegelgerade,GridVisible=TRUE,
ViewingBox=[-5..7,-5..5])
```



##

Dastellung der affinen Verformung

```
Eh:=matrix([[1,0],[0,1]])
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
As:=(1-s)*Eh+s*A
```

$$\begin{pmatrix} 1-\frac{s\cdot 2}{5} & -\frac{4\cdot s}{5} \\ -\frac{4\cdot s}{5} & 1-\frac{s\cdot 8}{5} \end{pmatrix}$$

```
myBMs:=As*myUrM
```

$$\begin{pmatrix} 2-\frac{s\cdot 4}{5} & 2-\frac{s\cdot 8}{5} & 1-\frac{s\cdot 6}{5} & 1-\frac{s\cdot 14}{5} & -\frac{12\cdot s}{5} & 0 & 2-\frac{s\cdot 4}{5} & 1-\frac{s\cdot 4}{5} \\ -\frac{8\cdot s}{5} & 1-\frac{s\cdot 16}{5} & 1-\frac{s\cdot 12}{5} & 3-\frac{s\cdot 28}{5} & 3-\frac{s\cdot 24}{5} & 0 & -\frac{8\cdot s}{5} & \frac{1}{2}-\frac{s\cdot 8}{5} \end{pmatrix}$$


```
myBs:=[[myBMs[1,j],myBMs[2,j]] $ j=1..npk]:
bilds:=plot::Polygon2d(myBs,s=0..1,

LineWidth=1, LineColor=RGB::Red,
```

```
LineWidth=1, LineColor=RGB::Red,
AnimationStyle=BackAndForth):
plot(urbild,bilds,e1s,e2s,spiegelgerade,GridVisible=TRUE,
ViewingBox=[-5..4,-5..3.5])
```



 animieren durch Anklicken!